

## Функция ПРОСМОТРХ и динамические массивы в Excel

Это перевод главы книги Bill Jelen. Excel Dynamic Arrays Straight to the Point. К [содержанию](#).

*ПРОСМОТРХ – первая новая функция после появления динамических массивов*

Функции ПРОСМОТРХ и ПОИСКПОЗХ дебютировали для Office 365 в августе 2019 года. Они были разработаны для объединения возможностей ВПР, ИНДЕКС, ПОИСКПОЗ и динамических массивов.

[Ранее](#) я писал о двенадцати удивительных преимуществах ПРОСМОТРХ.<sup>1</sup> Здесь расскажу лишь о двух из них. Но сначала введение для тех, кто еще не знаком с ПРОСМОТРХ.

В ячейке D4 вы хотите найти категорию (Category) для элемента W25-6. До появления ПРОСМОТРХ, вы могли это сделать с помощью =ВПР(A4;F6:I17;4;ЛОЖЬ) или комбинации функций =ИНДЕКС(I7:I17;ПОИСКПОЗ(A4;F7:F17;0)). С ПРОСМОТРХ сделать это еще проще. Вы ищете значение ячейки A4 в диапазоне F\$7:F\$17, и хотите вернуть соответствующее значение из I\$7:I\$17. Нет необходимости в аргументе ЛОЖЬ функции ВПР или 0 функции ИНДЕКС, потому что ПРОСМОТРХ по умолчанию ищет точное совпадение:

Formula bar: =ПРОСМОТРХ(A4;F\$7:F\$17;I\$7:I\$17)

Item	Date	Qty	Category
W25-6	01.08.2019	878	SKU: W25-6, Description: 18K Italian Gold Wom, Price: 69.95, Category: Wrist
CR 50-4	01.08.2019	213	Charm
CR 50-4	02.08.2019	744	Charm
BR26-3	03.08.2019	169	Wrist
CR50-6	03.08.2019	822	Charm
ER46-14	03.08.2019	740	Ear
RG78-25	03.08.2019	638	Ring
BR15-3	04.08.2019	817	Wrist
Cross50-5	04.08.2019	871	Charm
CR50-2	04.08.2019	686	Charm
BR26-3	05.08.2019	541	Wrist
ER41-4	05.08.2019	208	Ear
RG75-3	05.08.2019	231	Ring
RG75-3	05.08.2019	304	Ring
CR 50-4	05.08.2019	671	Charm
BR26-3	06.08.2019	849	Wrist
CR 50-4	06.08.2019	718	Charm

Рис. 68. Простой пример ПРОСМОТРХ

*Возврат 12 столбцов с помощью ВПР*

Если вам нужно вернуть значения для каждого из 12 месяцев, вы аккуратно используете смешанные ссылки, а также функцию СТОЛБЕЦ(), чтобы задать правильное значение третьего аргумента функции ВПР: =ВПР(\$A3;\$O\$3:\$AA\$226;СТОЛБЕЦ(В:В);ЛОЖЬ)

Formula bar: =ВПР(\$A3;\$O\$3:\$AA\$226;СТОЛБЕЦ(В:В);ЛОЖЬ)

Acct	Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec	J	F	M	A	M	J	J	A	S	O	N	D	
A308	6	1	9	2	9	1	5	4	4	4	7	3	A101	5	2	0	8	3	1	1	0	6	1	4	7
A219	0	0	5	0	2	8	8	7	8	3	0	0	A102	3	0	7	0	4	9	4	3	5	3	7	1
A249	1	7	1	3	2	9	6	7	1	2	9	5	A103	1	4	9	0	3	4	3	8	0	5	6	
A154	2	5	8	0	3	3	5	5	2	8	9	3	A104	0	4	2	5	0	0	9	7	6	8	8	5
A128	4	1	8	6	2	5	7	8	4	0	9	5	A105	9	4	8	1	3	4	5	7	7	7	9	7
A229	5	5	4	7	8	1	5	6	9	2	7	3	A106	6	3	6	2	6	2	1	1	3	8	7	5

Рис. 69. Хитрости использования ВПР для извлечения значений в несколько столбцов

<sup>1</sup> Рассказ от первого лица – автора книги Билла Джелена. – Прим. Багузина

### Возврат 12 столбцов с помощью ИНДЕКС и ПОИСКПОЗ

До появления динамических массивов, возможно, лучшим решением было использование вспомогательного столбца для ПОИСПОЗ и двенадцати столбцов для ИНДЕКС. Добавьте столбец В. Используйте заголовок *Где?* В ячейку В6 введите формулу =ПОИСКПОЗ(А6;\$Q\$6:\$Q\$229;0). Формула вернет номер той строки в таблице поиска, где находится А308. В ячейку С6 введите =ИНДЕКС(R\$6:R\$229;\$B6). Обратите внимание на относительные ссылки в первом и втором аргументах. При копировании вправо результат за январь автоматически изменится на февраль и т.д.

B6																													
=ПОИСКПОЗ(А6;\$Q\$6:\$Q\$229;0)																													
	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X	Y	Z	AA	AB	AC
1																													
2																													
3																													
4																													
5	Acct	Где?	Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec		Acct	J	F	M	A	M	J	J	A	S	O	N	D	
6	A308	208	6	1	9	2	9	1	5	4	4	4	7	3		A101	5	2	0	8	3	1	1	0	6	1	4	7	
7	A219															A102	3	0	7	0	4	9	4	3	5	3	7	1	
8	A249															A103	1	4	9	0	3	4	3	3	8	0	5	6	
9	A154															A104	0	4	2	5	0	0	9	7	6	8	8	5	
10	A128															A105	9	4	8	1	3	4	5	7	7	7	9	7	
11	A229															A106	6	3	6	2	6	2	1	1	3	8	7	5	
12	A111															A107	7	6	4	0	9	4	7	4	3	7	3	1	

Рис. 70. Вспомогательный столбец для ПОИСКПОЗ

### С динамическими массивами одна формула ПРОСМОТРХ вернет все значения

В ячейке В6 введите формулу =ПРОСМОТРХ(А6;\$O\$6:\$O\$229;\$P\$6:\$AA\$229). Она вернет 12 значений в диапазон В6:М6. Протяните формулу вниз вдоль столбца В.

B6																											
=ПРОСМОТРХ(А6;\$O\$6:\$O\$229;\$P\$6:\$AA\$229)																											
	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P	Q	R	S	T	U	V	W	X	Y	Z	AA
1																											
2																											
3																											
4																											
5	Acct	Jan	Feb	Mar	Apr	May	Jun	Jul	Aug	Sep	Oct	Nov	Dec		Acct	J	F	M	A	M	J	J	A	S	O	N	D
6	A308	6	1	9	2	9	1	5	4	4	4	7	3		A101	5	2	0	8	3	1	1	0	6	1	4	7
7	A219	0	0	5	0	2	8	8	7	8	3	0	0		A102	3	0	7	0	4	9	4	3	5	3	7	1
8	A249	1	7	1	3	2	9	6	7	1	2	9	5		A103	1	4	9	0	3	4	3	3	8	0	5	6
9	A154	2	5	8	0	3	3	5	5	2	8	9	3		A104	0	4	2	5	0	0	9	7	6	8	8	5
10	A128	4	1	8	6	2	5	7	8	4	0	9	5		A105	9	4	8	1	3	4	5	7	7	7	9	7
11	A229	5	5	4	7	8	1	5	6	9	2	7	3		A106	6	3	6	2	6	2	1	1	3	8	7	5
12	A111	0	7	1	1	5	8	2	5	2	2	6	2		A107	7	6	4	0	9	4	7	4	3	7	3	1

Рис. 71. Функция ПРОСМОТРХ с использованием динамических массивов

### Ошибка копирования формулы динамических массивов

Обычно, если вы введете формулу в В6, выберите В6 и дважды щелкните маркер заполнения, Excel скопирует формулу во все строки столбца В. Из-за ошибки это не работает, если В6 содержит динамический массив. (Ошибка все еще актуальна по состоянию на 25 сентября 2020 года.)

Обходной путь:

1. Из ячейки В6 нажмите клавишу со стрелкой влево, чтобы перейти к ячейке А6.
2. Из А6 нажмите Ctrl + стрелка вниз, чтобы перейти к последней используемой ячейке в столбце А.
3. Нажмите клавишу со стрелкой вправо, чтобы вернуться в столбец В.
4. Нажмите Ctrl + Shift + стрелка вверх, чтобы выбрать все ячейки столбца В вплоть до ячейки В6.
5. Нажмите Ctrl + D, чтобы скопировать формулу в верхней части выделения во все ячейки выделения.



### Двухкоординатный поиск с ИНДЕКС и ПОИСКПОЗ

Выполнение двухкоординатного поиска – еще один сценарий, в котором ИНДЕКС + ПОИСКПОЗ выигрывают у ВПР.<sup>2</sup> Рассмотрим следующий пример. Пользователь может выбрать счет в В3 и месяц в В4. Нужна формула, чтобы найти значение на пересечении счета А621 и месяца май.

На первом шаге ищем А621 в диапазоне номеров счетов. =ПОИСКПОЗ(В3;А9:А24;0) возвращает 12, т.е., А621 находится в 12-м ряду номеров счетов. Далее, =ПОИСКПОЗ(В4;В8:G8;0) возвращает 5, т.е., май – пятый месяц. Теперь, вы знаете, что вам нужна 12-я строка и 5-й столбец массива В9:G21: =ИНДЕКС(В9:G24;ПОИСКПОЗ(В3;А\$9:А\$24;0);ПОИСКПОЗ(В4;В\$8:G\$8;0))

	A	B	C	D	E	F	G	H	I	J	K	L	M
1													
2													
3	Acct:	A621		12	=ПОИСКПОЗ(В3;А9:А24;0)								
4	Month:	May		5	=ПОИСКПОЗ(В4;В8:G8;0)								
6	Result?:	:0)			=ИНДЕКС(В9:G24;ПОИСКПОЗ(В3;А\$9:А\$24;0);ПОИСКПОЗ(В4;В\$8:G\$8;0))								
7													
8	Acct	Jan	Feb	Mar	Apr	May	Jun						
9	A302	99	41	31	114	90	1						
10	A397	10	56	3	7	44	50						
11	A806	49	71	83	37	110	14						
12	A936	2	101	18	63	54	95						
13	A740	98	20	80	59	96	16						
14	A945	36	19	106	52	35	70						
15	A695	62	43	58	34	51	69						
16	A571	13	108	27	67	86	68						
17	A315	11	9	4	76	30	23						
18	A552	55	26	28	85	111	81						
19	A628	40	87	88	47	107	60						
20	A621	74	57	103	33	104	64						
21	A211	93	92	84	61	112	79						
22	A563	53	15	8	42	24	46						
23	A339	25	72	73	113	21	6						
24	A673	17	89	82	65	5	109						

Рис. 72. ИНДЕКС + ПОИСПОЗ для двухкоординатного поиска

### Двухкоординатный поиск с ПРОСМОТРХ

Формула на основе функции ПРОСМОТРХ будет короче, но... сложнее для понимания. Итак, вы хотите найти А621 в списке учетных записей, поэтому формула будет начинаться с =ПРОСМОТРХ(В3;А12:А27;...). Но теперь самое сложное. Как вам получить аргумент *просматриваемый\_массив*?

Вы используете тот же трюк, который применили для возврата 12 месяцев, но транспонированный. В следующей формуле ПРОСМОТРХ выполняет горизонтальный поиск: =ПРОСМОТРХ(В4;В11:G11;В12:G27). Обычно вы используете одну строку в качестве аргумента *просматриваемый\_массив*. И ПРОСМОТРХ возвращает одно значение. Если же в качестве аргумента *просматриваемый\_массив* ввести прямоугольный диапазон, ПРОСМОТРХ вернет один столбец из 16 строк.

Этот промежуточный результат можно увидеть диапазоне I12:I27. Если в ячейке В4 изменить май на февраль, то формула вернет числа из столбца С12:С27. Трюк состоит в том, чтобы использовать внутренний ПРОСМОТРХ для получения значений, подставляемых в аргумент *просматриваемый\_массив* внешнего ПРОСМОТРХ.

<sup>2</sup> Не знаю, насколько выигрывает... Вот формула с ВПР, которая решает эту задачу: =ВПР(В3;А8:G24;ПОИСКПОЗ(В4;А8:G8;0);ЛОЖЬ)

B7		=ПРОСМОТРХ(В3;А12:А27;ПРОСМОТРХ(В4;В11:G11;В12:G27))												
	A	B	C	D	E	F	G	H	I	J	K	L	M	
1														
3	<b>Acct:</b>	A621												
4	<b>Month:</b>	May												
6	<b>Result?</b>	104	=ИНДЕКС(В12:G27;ПОИСКПОЗ(В3;\$A\$12:\$A\$27;0);ПОИСКПОЗ(В4;\$B\$11:\$G\$11;0))											
7	<b>Result?</b>	104	=ПРОСМОТРХ(В3;А12:А27;ПРОСМОТРХ(В4;В11:G11;В12:G27))											
9														
10													Почему это работает?	
11	<b>Acct</b>	<b>Jan</b>	<b>Feb</b>	<b>Mar</b>	<b>Apr</b>	<b>May</b>	<b>Jun</b>						Внутренний ПРОСМОТРХ возвращает	
12	A302	99	41	31	114	90	1						столбец результатов:	
13	A397	10	56	3	7	44	50						90 =ПРОСМОТРХ(В4;В11:G11;В12:G27)	
14	A806	49	71	83	37	110	14						44	
15	A936	2	101	18	63	54	95						110	
16	A740	98	20	80	59	96	16						54	
17	A945	36	19	106	52	35	70						96	
18	A695	62	43	58	34	51	69						35	
19	A571	13	108	27	67	86	68						51	
20	A315	11	9	4	76	30	23						86	
21	A552	55	26	28	85	111	81						30	
22	A628	40	87	88	47	107	60						111	
23	A621	74	57	103	33	104	64						107	
24	A211	93	92	84	61	112	79						104	
25	A563	53	15	8	42	24	46						112	
26	A339	25	72	73	113	21	6						24	
27	A673	17	89	82	65	5	109						21	

Рис. 73. Формула на основе ПРОСМОТРХ короче, но сложнее